

Problem 69 with PVS

Kai Engelhardt

January 24, 2023

1 A No-Frills Encoding

```
1  % Problem number "69. Greatest Common Divisor Algorithm from the list
2  % of the "top 100" of mathematical theorems - which nowadays serves as
3  % some form of benchmark for theorem provers at
4  % <https://www.cs.ru.nl/~freek/100/> - lacked a PVS formalisation
5  % and proof.
6
7  gcd2: THEORY
8
9  BEGIN
10 x,y,z: VAR posnat
11
12 % axiomatic gcd, inspired by https://isabelle.in.tum.de/dist/library/HOL/HOL/GCD.html
13 gcd(x,y): posnat
14 gcd_dvd1: AXIOM divides(gcd(x,y),x)
15 gcd_dvd2: AXIOM divides(gcd(x,y),y)
16 gcd_greatest: AXIOM divides(z,x) IMPLIES divides(z,y) IMPLIES divides(z,gcd(x,y))
17
18 % some simple properties that ought to follow
19 gcd_eq: LEMMA gcd(x,x) = x
20 gcd_sym: LEMMA gcd(x,y) = gcd(y,x)
21 gcd_stp: LEMMA x > y IMPLIES gcd(x,y) = gcd(x-y,y)
22 gcd_stp2: LEMMA x < y IMPLIES gcd(x,y) = gcd(x,y-x)
23 gcd_leq: LEMMA x >= gcd(x,y)
24
25 % algorithmic gcd, Euclid's algorithm
26 gcd2(x,y): RECURSIVE posnat =
27   TABLE
28     %-----+-----+-----+
29     | [ x = y | x > y          | ELSE          ] |
30     %-----+-----+-----+
31     | x          | gcd2(x-y,y) | gcd2(x,y-x) ||
32     %-----+-----+-----+
33   ENDTABLE
34   MEASURE x+y
35
36 gcd2_h1: LEMMA x > y IMPLIES gcd2(x-y,y) = gcd(x-y,y) IMPLIES gcd2(x,y) = gcd(x,y)
37 gcd2_h2: LEMMA x < y IMPLIES gcd2(x,y-x) = gcd(x,y-x) IMPLIES gcd2(x,y) = gcd(x,y)
38
39 gcd2_eq_gcd: THEOREM gcd2(x,y) = gcd(x,y)
40
41 END gcd2
```

The proofs are straightforward und uninspiring, resulting in the summary:

Proof summary for theory gcd2

gcd_eq.....	proved - complete	[shostak] (0.04 s)
gcd_sym.....	proved - complete	[shostak] (0.01 s)
gcd_stp_TCC1.....	proved - complete	[shostak] (0.02 s)
gcd_stp.....	proved - complete	[shostak] (0.08 s)
gcd_stp2_TCC1.....	proved - complete	[shostak] (0.02 s)
gcd_stp2.....	proved - complete	[shostak] (0.04 s)
gcd_leq.....	proved - complete	[shostak] (0.01 s)
gcd2_TCC1.....	proved - complete	[shostak] (0.01 s)
gcd2_TCC2.....	proved - complete	[shostak] (0.03 s)
gcd2_TCC3.....	proved - complete	[shostak] (0.00 s)
gcd2_TCC4.....	proved - complete	[shostak] (0.01 s)
gcd2_h1.....	proved - complete	[shostak] (0.02 s)
gcd2_h2.....	proved - complete	[shostak] (0.02 s)
gcd2_eq_gcd.....	proved - complete	[shostak] (0.23 s)
Theory gcd2 totals: 14 formulas, 14 attempted, 14 succeeded		(0.54 s)